

Paper 4653-2020

Obtaining National Readmission Estimates: Examining Surgical Readmission Rates in the United States

Patrick Karabon, Alexander Balinski, and Rachel Patterson, Oakland University
William Beaumont School of Medicine

ABSTRACT

In the past several years, there has been increased focus from healthcare stakeholders in reducing the rate of inpatient hospital readmissions. Policies, including the Hospital Readmissions and Reduction Program (HRRP) from the Centers of Medicare and Medicaid Services (CMS), have placed financial penalties on hospitals with excessive readmission rates. The Nationwide Readmissions Database (NRD) is a comprehensive nationwide database that enables users to obtain population-based national estimates of readmissions. With a claims-based data structure, this database allows for analysis of readmission trends, analysis of primary cause of readmission, or even analysis of time to readmission.

This E-Poster shows how to obtain population-based estimates and trends by using both PROC SQL and the SAS/STAT® Survey Sampling and Analysis Procedures. Emphasis was placed on working with the claims-based data structure of the NRD and obtaining accurate estimates of standard errors for readmission rates. Examples involve readmissions in thyroidectomies, one of the most commonly performed procedures, as well as Transcatheter Aortic Valve Replacement (TAVR), a newly developed approach to heart valve replacement.

INTRODUCTION

In the last decade, the healthcare sector has placed an increased focus on reducing hospital readmissions who have recently been discharged from a surgical procedure. Many of these stakeholders see inpatient readmissions as a high priority healthcare quality measure. As of 2016, nearly 14 out of every 100 hospitalizations in the United States is readmitted to a hospital within 30 days of the initial discharge. Readmissions also shoulder a large economic burden because, on average, each readmission also incurs total charges of approximately \$14,400 based on 2016 data (Bailey et al., 2019).

With readmissions contributing a large burden on the United States healthcare system, recent programs like the Hospital Readmissions and Reduction Program (HRRP) have encouraged lower readmission rates across the United States. Even patients are now empowered, as they can look up 30-day readmission rates due to complications for their surgeon and hospital for a variety of elective surgical procedures through the ProPublica Surgeon Scorecard (Wei et al., 2015).

THE HOSPITAL READMISSIONS AND REDUCTION PROGRAM (HRRP)

The 30-day readmission rate for Medicare patients was over 18% in 2010 and 2011, which was a higher rate than patients who carry Medicaid, Private Insurance, or were even uninsured (Bailey et al., 2019). Established in 2012 as part of the Patient Protection and Affordable Care Act (PPACA), the HRRP was developed to promote reducing high readmission rates in the Medicare population. In the HRRP program, hospitals that are determined to have excessive readmission for specified conditions, as calculated by an algorithm, can have Medicare reimbursements penalized by up to 3% for the next year. The program promotes increased quality of care through this penalty to hospitals who have a high readmission rate (Centers for Medicare and Medicaid Services, 2018).

TAVR AND THYROIDECTOMY

This E-Poster will focus on 30-day readmission rates for two surgical procedures: Transcatheter Aortic Valve Replacement (TAVR) and Thyroidectomy.

TAVR is a minimally invasive procedure used to replace the aortic valve in patients with severe hardening of the aortic valve, which is a situation that can contribute subsequent Congestive Heart Failure. In lieu of intensive open heart surgery to replace the aortic valve, TAVR is preferred in patients who are at high risk of death or complications and thus cannot tolerate an intensive open heart surgery (Rodes-Cabau, 2012). Due to the procedure being favored for patients with many comorbidities or high risk of complications or death, some studies have found that TAVR readmission rates were no better or worse than readmission rates for traditional open heart surgery (Vejjongsakul et al., 2017; Dhoble et al., 2018). Therefore, measuring TAVR readmission rates is important to understand the current state of readmissions for TAVR.

Thyroidectomy is a very common surgical procedure and involves the removal of part (partial thyroidectomy) or all (complete thyroidectomy) of the thyroid gland. There are many reasons to remove a thyroid gland, including surgery or other endocrine conditions that necessitate thyroid removal. While Thyroidectomy represents a relatively routine surgery for many surgeons, subsequent complications can still occur for a wide variety of reasons (Brunnicardi, 2019).

THE NATIONWIDE READMISSIONS DATABASE (NRD)

The Nationwide Readmission Database (NRD) is a database created and managed by the Agency for Healthcare Research and Quality (AHRQ) and contains data on approximately 18 million hospital discharges per calendar year, which are contributed from hospitals located in 28 different states in the United States. The main purpose of analyzing this database is to obtain population-based readmission estimates. Further types of analyses are possible and can include analysis of causes of readmission, time to readmission from initial discharge, and even costs and other outcomes related to admissions (Agency for Healthcare Research and Quality, 2018). While resembling claims-based data, the NRD is unique because of its complex sampling design, which means that SAS/STAT® Survey Sampling and Analysis Procedures are used for data analysis.

A unique de-identified patient linkage variable (NRD_VISITLINK) is included in the NRD and is used to identify both the index event (the original hospitalization where the surgical procedure took place) and any subsequent readmissions that occur within an analyst-specified time window (Agency for Healthcare Research and Quality, 2019). While this E-Poster concentrates on 30-day readmission rates, analysts can identify readmissions for time periods of up to 90 days after discharge from the index event (Yoon et al., 2017).

COMPLEX SAMPLING DESIGN OF THE NRD

Since the NRD has a complex sampling design, SAS/STAT® Survey Sampling and Analysis procedures are required for analysis and this means that there are additional variables that are included within the NRD database in order to conduct a proper analysis. Since only 28 states contribute data to the NRD and not all hospitals in these states are included, a discharge weight (DISCWT) is created in order to correct for over or under representation of discharges relative to all hospital discharge in the United States. In order to properly calculate the standard errors, the NRD also includes a Cluster variable (HOSP_NRD) and a Strata variable (NRD_STRATUM). The three aforementioned variables are included in all Survey Analysis procedures (Yoon et al., 2017).

SUBPOPULATION/DOMAIN ANALYSIS IN THE NRD

In many scenarios, including the examples used in this E-Poster, require that a subpopulation or domain analysis is run. For example, there are two important exclusions that we will make from the NRD to run the analysis: excluding pediatric cases (cases performed in patients < 18 years of age) and cases who died during the initial hospitalization. Furthermore, since we are looking at 30-day readmissions, we exclude procedures where the discharge was done in December because these cases will not have a 30-day readmission window. Unlike most sources of claims-based data, each year of the NRD is treated as a separate sample (Yoon et al., 2017).

Since a subpopulation, or domain analysis, is required, the DOMAIN statement is necessary in all of the Analysis procedures. The Illustrative Examples include code to show how to create the subpopulation variable.

ILLUSTRATIVE EXAMPLES

EXAMPLE #1: TAVR READMISSION RATE

For this first illustrative example, we are going to obtain a population-based 30-day readmission rate for TAVR using the 2014 NRD database. The code used to identify cases is modified from Yoon et al. (2017), where modifications are added to the code in order to make it flexible and identify cases for TAVR, which is the surgical procedure of interest. In addition, the code block can be modified for various types of analysis and different types of exclusions, as specified by the analyst and the research team.

The first step is to create a dataset that contains all of the TAVR index events, or discharges where TAVR was the primary procedure performed during that hospitalization. Note that in this code, there are three macro variables created using %LET statements to identify the cases of interest. The first %LET statement is the year of data while the second %LET statement identifies the ICD-9-CM Procedure Codes 35.05 and 35.06 that are used to search the NRD for TAVR cases. Finally, the timeframe is set to 30 which indicates that we are interested in 30-day readmission rates. This code block, modified from the Yoon et al. (2017) example code is as follows:

```
%LET YEAR = 2014;
%LET CODES = "3505", "3506";
%LET TIMEFRAME = 30;
DATA NRD_&YEAR._INDEXEVENTS
  READMCDATAS (DROP = DISCWT LOS NRD_STRATUM INDEXEVENT);
  SET NRD_&YEAR._CORE (OBS = MAX
    KEEP = HOSP_NRD KEY_NRD DISCWT NRD_STRATUM AGE
    DMONTH DIED LOS NRD_VISITLINK
    NRD_DAYSTOEVENT PR1);
  ATTRIB INDEXEVENT LENGTH = 3 LABEL = "TAVR Index Event";
  INDEXEVENT = 0;

  /* Inclusion/Exclusion Criteria to Identify TAVR Index Events */
  IF (PR1 IN (&CODES.)
    AND AGE GE 18
    AND DMONTH LE 11
    AND DIED = 0
    AND NOT MISSING(LOS)
  THEN INDEXEVENT = 1;
  DROP PR1 AGE MONTH DIED;
  IF INDEXEVENT = 1 THEN OUTPUT NRD_&YEAR._INDEXEVENTS;
  OUTPUT READMCDATAS;

RUN;
```

After identifying the TAVR index events in the previous DATA step, the next step is to identify all of the readmissions that occurred within 30 days of discharge from the initial TAVR case. Readmissions are found using PROC SQL and this is also modified from the code provided by Yoon et al. (2017) to fit the specific needs of our project. The code is as follows:

```

PROC SQL;
CREATE TABLE READMISSIONSALL AS
SELECT I.HOSP_NRD AS HOSP_NRD_INDEX
      , I.KEY_NRD AS KEY_NRD_INDEX
      , R.*
FROM NRD_&YEAR._INDEXEVENTS I
INNER JOIN READMCANDIDATES R
ON I.NRD_VISITLINK = R.NRD_VISITLINK
AND I.KEY_NRD NE R.KEY_NRD
AND R.NRD_DAYSTOEVENT - (I.NRD_DAYSTOEVENT + I.LOS) BETWEEN 0
AND &TIMEFRAME.
AND I.INDEXEVENT = 1;
ORDER BY I.HOSP_NRD, I.KEY_NRD, R.NRD_DAYSTOEVENT;
QUIT;

```

There are a few subsequent DATA steps that are necessary. One important step is deciding whether only the first readmission is necessary or if data from all readmissions is required to complete your analysis. In this example (and the subsequent example), we only want data for the first readmission. The full code is available or Yoon et al. (2017) can be consulted to see what the code should look like; however, using either method, the code will have to be modified to meet the specific needs of the project.

Based on the code above, the Table 1 below shows the various numbers (all unweighted) that are obtained through this process:

	Unweighted Cases
All cases in CORE File/Readmission Candidates	14,894,613
TAVR Index Events (Procedures)	879
All Readmission Events Related to TAVR	143
First Readmission Events Only	126

Table 1: TAVR Cases in the 2014 NRD

The algorithm first looks at the 14,894,613 cases that are contained in the 2014 CORE dataset, which represents all of the cases in the 2014 NRD. Of the cases in the CORE dataset, 879 meet the definition for TAVR by having a primary ICD-9 Procedure Code of 35.05 or 35.06 and meet the other inclusion criteria. Of these 879 TAVR cases, the PROC SQL step finds 143 readmissions within 30 days for these TAVR cases. After excluding multiple readmissions because the interest is the first readmission only, the algorithm finds that 126 patients are readmitted within 30 days of discharge.

Now that the data is prepared for analysis, the following PROC SURVEYMEANS code is run to obtain an estimate of the 30-day readmission rate. Note the use of the WEIGHT, CLUSTER, STRATA, and DOMAIN statements to run the appropriate analysis, including accurate standard errors. The code to obtain the population-based readmission rate is as follows:

```

PROC SURVEYMEANS DATA = TAVR_2014 SUMWGT SUM MEAN PLOTS = NONE;
CLUSTER HOSP_NRD;
STRATA NRD_STRATUM;
WEIGHT DISCWT;
VAR READMIT;
DOMAIN INSUBSET*INDEXEVENT;
RUN;

```

From the previous PROC SURVEYMEANS code, the following 30-day readmission rate with 95% Confidence Limits is shown in Table 2 below:

30-Day Readmission Rate	95% Confidence Limits	
	Lower Bound	Upper Bound
14.33%	11.76%	16.90%

Table 2: TAVR 30-Day Readmission Rate Estimate

From Table 2 above, 14.33% of TAVR cases were readmitted to the hospital within 30 days of discharge as based on the 2014 NRD data. In graphical form using PROC SGPLOT, the results for the population-based readmission rate with 95% Confidence Limits appears as follows in Figure 1:




Figure 1: TAVR 30-Day Readmission Rate Estimate

EXAMPLE #2: THYROIDECTOMY READMISSION TREND

The Clinical Classification System (CCS), developed by AHRO, is a system that combines ICD-9 Diagnosis and Procedure codes into meaningful Diagnosis and Procedure groups (Healthcare Cost and Utilization Project, 2016). Of the CCS Procedure code groups, Thyroidectomy has a CCS code assigned to it, in the latest version, a CCS Procedure group of 10 indicates a Thyroidectomy. Much like the PR variables to delineate individual codes, there are PRCSS variables that are used to identify cases through the CCS. While not shown in the paper, the code from Example #1 is modified again to find Thyroidectomy cases.

As the name of the example implies, we are interested in looking at the trend in the 30-day readmission rate for Thyroidectomy cases between 2011 and 2014. Since each year needs to be treated as a separate sample, year is considered a categorical variable (Yoon et al., 2017). Since the outcome variable, 30-day readmissions, is also a categorical variable, the Cochran-Armitage Test for Trend makes the most sense; however, this test is not possible when working with complex samples data. Therefore, the association between year and 30-day readmission is examined using a Rao-Scott Chi-Square test, which is an extension of the Chi-Square test that adjusts for the complex sampling design variables contained in the

NRD. In addition, a Complex Samples Logistic Regression is run as an additional analysis to ensure the trend analysis results are robust.

Before running any trend analysis, it is good practice to calculate population-based rates of 30-day readmissions for every year of the data. Recall that every year of the NRD is treated as a separate sample; therefore, YEAR is included in the DMOAIN statement. The readmission rates are calculated using the following PROC SURVEYMEANS code:

```
PROC SURVEYMEANS DATA = THYROID SUMWGT SUM MEAN PLOTS = NONE ;
  CLUSTER HOSP_NRD ;
  STRATA NRD_STRATUM ;
  WEIGHT DISCWT ;
  VAR READMIT ;
  DOMAIN INSUBSET*INDEXEVENT*YEAR ;
RUN ;
```

From the previous PROC SURVEYMEANS code, the Table 3 shows the 30-day readmission rates for Thyroidectomies with 95% Confidence Limits:

Year of Procedure	Number of Procedures	30-Day Readmission Rate	95% Confidence Limits	
			Lower Bound	Upper Bound
2011	48,562	5.45%	4.92%	5.98%
2012	42,362	5.36%	4.91%	5.81%
2013	35,203	5.44%	4.99%	5.90%
2014	27,564	6.00%	5.48%	6.51%

Table 3: Thyroidectomy 30-Day Readmission Rate Estimates

The corresponding graphical output for the population-based readmission rate estimates from PROC SGPLOT is as follows in Figure 2:




Figure 2: Thyroidectomy 30-Day Readmission Rate Estimates

For the trend analysis, the objective is to determine whether there is a significant trend over time in the 30-day readmission rate for Thyroidectomies between 2011 and 2014. The first option is to run a Rao-Scott Chi-Square test to see if there is an association between year of

procedure and 30-day readmissions. The code to run the Rao-Scott Chi-Square test is as follows:

```
PROC SURVEYFREQ DATA = THYROID;
  CLUSTER HOSP_NRD;
  STRATA NRD_STRATUM;
  WEIGHT DISCWT;
  TABLES INSUBSET*READMIT*YEAR / CHISQ;
RUN;
```

From the above code, the corresponding Rao-Scott Chi-Square test output is as follows:

Pearson Chi-Square	7.1839
Design Correction	2.1651
Rao-Scott Chi-Square	3.3180
DF	3
Pr > ChiSq	0.3451

Table 4: Rao-Scott Chi-Square Test to Evaluate Trend in Thyroidectomy Readmissions

Based on the output above, it appears that there is not enough evidence to conclude that there is an association between 30-day readmissions and year; therefore, there does not appear to be a trend in Thyroidectomy readmissions between 2011 and 2014 (P = 0.3451).

Another way to look at the temporal trend is to use PROC SURVEYLOGISTIC. Note that the variable YEAR is treated as a categorical variable because NRD is repeated cross-sections or panel data and is not longitudinal. The code to run PROC SURVEYLOGISTIC is shown below:

```
PROC SURVEYLOGISTIC DATA = THYROID;
  CLUSTER HOSP_NRD;
  STRATA NRD_STRATUM;
  WEIGHT DISCWT;
  DOMAIN INSUBSET;
  CLASS READMIT(REF = "0") YEAR(REF = "2011") / PARAM = GLM;
  MODEL READMIT = YEAR;
RUN;
```

From the previous code block to evaluate the temporal trend, the Type 3 P-Value for the variable year is helpful:

Effect	F Value	Num DF	Den DF	Pr > F
YEAR	0.79	3	7226	0.5006

Table 5: Logistic Regression Type 3 P-Value to Evaluate Trend in Thyroidectomy Readmissions

From this above table, there does not appear to be enough evidence that there is a significant trend in 30-day readmission rates over time for Thyroidectomy (P = 0.5006). In addition, the parameter estimates from PROC SURVEYLOGISTIC are shown below in Table 6 to see if there is any trend in the odds of 30-day readmissions over the study period:

Parameter		Odds Ratio	95% CL Lower Bound	95% CL Upper Bound	P-Value
Intercept					< 0.0001
YEAR	2012	0.98	0.86	1.12	0.7691
YEAR	2013	0.98	0.86	1.12	0.7632
YEAR	2014	1.07	0.93	1.22	0.3537
YEAR	2011				

Table 6: Logistic Regression Parameter Estimates

From Table 6 above, there does not appear that there is enough evidence to conclude that there is a significant difference in 30-day readmission rates for Thyroidectomy in any year between 2011 and 2014 (all $P \geq 0.0500$).

CONCLUSION

Proper analysis of hospital readmissions in the United States is important as a wide variety of stakeholders in healthcare are interested in hospital readmissions as a quality of care measure. The DATA step and PROC SQL help Users quickly identify surgical cases of interest as well as subsequent readmissions for patients undergoing surgical procedures in the claims-based data structure of the NRD. Through the two illustrative examples, the SAS/STAT® Survey Sampling and Analysis Procedures are employed to obtain population-based readmission rates from the NRD.

REFERENCES

- Agency for Healthcare Research and Quality. 2019. "Overview of the Nationwide Readmissions Database." Accessed January 2020. <https://hcup-us.ahrq.gov/nrdoverview.jsp>.
- Bailey, M. K., Weiss, A. J., Barrett, M. L., and Jiang, H. J. 2019. "Statistical Brief #248: Characteristics of 30-Day All-Cause Hospital Readmissions, 2010-2016." Accessed January 2020. <https://www.hcup-us.ahrq.gov/reports/statbriefs/sb248-Hospital-Readmissions-2010-2016.jsp>.
- Brunnicardi, F. and et al. eds. 2019. *Schwartz's Principles of Surgery, 11th edition*. New York, NY: McGraw-Hill.
- Centers for Medicare and Medicaid Services. 2018. "Hospital Readmission and Reduction Program." Accessed January 2020. <https://www.cms.gov/Medicare/Quality-Initiatives-Patient-Assessment-Instruments/Value-Based-Programs/HRRP/Hospital-Readmission-Reduction-Program>.
- Dhoble, A., Bhise, V., Nevah, M. I. and et al. 2018. "Outcomes and readmissions after transcatheter aortic valve replacement in patients with cirrhosis: A propensity matched analysis." *Catheter Cardiovascular Interventions*, 91(1): 90-96.
- Healthcare Cost and Utilization Project. 2016. "Clinical Classifications Software (CCS) 2015." Accessed January 2020. <https://www.hcup-us.ahrq.gov/toolssoftware/ccs/CCSUsersGuide.pdf>.
- Rodes-Cabau, J. 2012. "Transcatheter aortic valve implantation: current and future approaches." *Nature Reviews Cardiology*, 9(1): 15-29.
- Vejpongsa, P., Bhise, V., Charitakis, K., and et al. 2017. "Early readmissions after transcatheter and surgical aortic valve replacement." *Catheter Cardiovascular Interventions*, 90(4): 662-670.
- Wei, S., Pierce, O., and Allen, M. 2015. "Surgeon Scorecard." Accessed January 2020. <https://projects.propublica.org/surgeons>.
- Yoon, F., Sheng, M., Jiang, H. J., Steiner, C. A., and Barrett, M. L. 2017. "Calculating Nationwide Readmission Database (NRD) Variances." Accessed January 2020. <https://www.hcup-us.ahrq.gov/reports/methods/2017-01.pdf>.

ACKNOWLEDGMENTS

The authors acknowledge the Agency for Healthcare Research and Quality (AHRO) Healthcare Cost and Utilization Project (HCUP) for the comprehensive collection of data on healthcare throughout the United States to use in their databases. The authors also want to thank those individuals at HCUP who provide structured code examples using SAS, which make analysis of the data much easier and ensure the accuracy of the results presented.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Patrick Karabon, MS
Oakland University William Beaumont School of Medicine
pkarabon@oakland.edu
www.oakland.edu/medicine